

**CHCDF****PURPOSE**

Compute the chi cumulative distribution function with degrees of freedom parameter  $v$ .

**DESCRIPTION**

The distribution of the positive square root of a variable having a chi-square distribution is a chi-distribution. The chi-distribution has the following probability density function:

$$f(x, v) = \frac{e^{-\frac{x^2}{2}} x^{v-1}}{2^{\frac{v}{2}-1} \Gamma\left(\frac{v}{2}\right)} \quad x > 0, v > 0 \quad (\text{EQ Aux-64})$$

where  $\Gamma$  is the gamma function. The input value and the degrees of freedom should both be positive real numbers.

The cumulative distribution is the area under the curve from 0 to  $x$  (i.e., the integral of the above function). It has the formula:

$$F(x, v) = \frac{\Gamma_a\left(\frac{v}{2}, \frac{x^2}{2}\right)}{\Gamma\left(\frac{v}{2}\right)} \quad x > 0, v > 0 \quad (\text{EQ Aux-65})$$

where  $\Gamma_a$  is the incomplete gamma function and  $\Gamma$  is the complete gamma function.

**SYNTAX**

LET <y2> = CHCDF(<y1>,<v>) <SUBSET/EXCEPT/FOR qualification>

where <y1> is a positive number, parameter, or variable;

<y2> is a variable or a parameter (depending on what <y1> is) where the computed chi cdf value is stored;

<v> is a positive number, parameter, or variable that specifies the degrees of freedom;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

**EXAMPLES**

LET A = CHCDF(3,10)

LET A = CHCDF(A1,10)

LET X2 = CHCDF(X1,10)

**NOTE**

The chi-distribution includes several distributions as special cases. If  $v$  is 1, the chi-distribution reduces to the half-normal distribution.

If  $v$  is 2, the chi-distribution is a Rayleigh distribution. If  $v$  is 3, the chi-distribution is a Maxwell-Boltzmann distribution. A generalized Rayleigh distribution is a chi-distribution with a scale parameter equal to 1.

**DEFAULT**

None

**SYNONYMS**

None

**RELATED COMMANDS**

CHPDF	=	Compute the chi probability density function.
CHPPF	=	Compute the chi percent point function.
CHSCDF	=	Compute the chi-square cumulative distribution function.
CHSPDF	=	Compute the chi-square probability density function.
CHSPPF	=	Compute the chi-square percent point function.
WEICDF	=	Compute the Weibull cumulative distribution function.
WEIPDF	=	Compute the Weibull probability density function.
WEIPPF	=	Compute the Weibull percent point function.
NORCDF	=	Compute the normal cumulative distribution function.

NORPDF = Compute the normal probability density function.  
 NORPPF = Compute the normal percent point function.

## REFERENCE

“Continuous Univariate Distributions,” Johnson, Kotz, and Balakrishnan, John Wiley and Sons, 1994, (chapter 18).

“Statistical Distributions,” 2nd ed., Evans, Hastings, and Peacock, John Wiley and Sons, 1993, (chapters 8 and 34).

## APPLICATIONS

Reliability

## IMPLEMENTATION DATE

95/4

## PROGRAM

```
LET STRING S1 = HALF-NORMAL DISTRIBUTION
LET STRING S2 = RAYLEIGH DISTRIBUTION
LET STRING S3 = MAXWELL-BOTZMAN DISTRIBUTION
LET STRING S4 = V=4
LET STRING S5 = V=5
LET STRING S6 = V=6
LET STRING S7 = V=7
LET STRING S8 = V=8
LET STRING S9 = V=9
MULTIPLY 3 3; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE AUTOMATIC
LOOP FOR K = 1 1 9
  X1LABEL ^S^K
  PLOT CHCDF(X,K) FOR X = 0.01 0.01 5
END OF LOOP
END OF MULTIPLY
```

